



HAL
open science

Formalising Sylow's theorems in Coq

Laurent Théry, Laurence Rideau

► **To cite this version:**

Laurent Théry, Laurence Rideau. Formalising Sylow's theorems in Coq. [Technical Report] 2006, pp.23. inria-00113750v1

HAL Id: inria-00113750

<https://inria.hal.science/inria-00113750v1>

Submitted on 14 Nov 2006 (v1), last revised 22 Nov 2006 (v2)

HAL is a multi-disciplinary open access archive for the deposit and dissemination of scientific research documents, whether they are published or not. The documents may come from teaching and research institutions in France or abroad, or from public or private research centers.

L'archive ouverte pluridisciplinaire **HAL**, est destinée au dépôt et à la diffusion de documents scientifiques de niveau recherche, publiés ou non, émanant des établissements d'enseignement et de recherche français ou étrangers, des laboratoires publics ou privés.



INSTITUT NATIONAL DE RECHERCHE EN INFORMATIQUE ET EN AUTOMATIQUE

Formalising Sylow's theorems in Coq

Laurence Rideau — Laurent Théry

N° ????

Novembre 2006

Thème SYM

 *Rapport
technique*



Formalising Sylow's theorems in Coq

Laurence Rideau , Laurent Théry

Thème SYM — Systèmes symboliques
Projet Marelle

Rapport technique n° ???? — Novembre 2006 — 23 pages

Abstract: This report presents a formalisation of Sylow's theorems done in COQ. The formalisation has been done in a couple of weeks on top of Georges Gonthier's SSREFLECT [2]. There were two ideas behind formalising Sylow's theorems. The first one was to get familiar with Georges way of doing proofs. The second one was to contribute to the collective effort to formalise a large subset of group theory in COQ with some non-trivial proofs.

Key-words: Group theory, Sylow's theorems, Formalisation of mathematics

Formalisation des théorèmes de Sylow dans Coq

Résumé : Ce rapport présente une formalisation des théorèmes de Sylow faite dans le système COQ. La formalisation s'est faite en deux semaines au dessus de la librairie SSREFLECT de Georges Gonthier. Il y avait deux principales motivations pour formaliser les théorèmes de Sylow. La première était de se familiariser avec la façon qu'a Georges de faire des preuves. La seconde était de contribuer à l'effort collectif de formaliser un large ensemble de la théorie des groupes en COQ.

Mots-clés : Théorie des groupes, Théorème de Sylow, Formalisation des mathématiques

1 Introduction

Sylow's theorems are central in group theory. Any course has a section or a chapter on them. Taking them as a first step in an effort to formalise group theory seemed a good idea. One of these theorems is number 72 in the list of the 100 theorems [4] maintained by Freek Wiedijk. Surprisingly, only one formalisation is known. It has been done in Isabelle by Florian Kammüller [3]. The proof that has been formalised in Isabelle is due to Wielandt [5]. It is a very concise and elegant proof. A central step in the proof is a non-trivial combinatorial argument that is used to show the existence of a group with a particular property. This is not the proof we have chosen to formalise. As we are interested in formalising Sylow's theorems not only as a mere exercise but as a base for further development, conciseness is nice but reusability is much more important. We have chosen to follow the proof given by Gregory Constantine [1] in his group theory course. It has the nice property of using one main tool, namely group actions, to prove most of the key results. The combinatorial argument that was present in the proof of Wielandt is then reduced to a minimum. Most of our formalising time has then been spent proving theorems about groups not about numbers.

The presentation of this work is organised as follows. In a first section, we describe what we started from. The main points we want to address are how SSREFLECT is organised and how using this dedicated version of COQ differs from using the standard one. In a second section, we outline the main steps of our proofs. Then, in a last section we conclude.

2 From types with decidable equality to finite types

2.1 Types with decidable equality

One of the key decision of SSREFLECT is to base the development on objects not in `Type` but in `eqType`, i.e objects for which equality is decidable.

```
Structure eqType : Type := EqType {
  sort  :> Set;
  eq    : sort -> sort -> bool;
  eqP   : forall x y, reflect (x = y) (eq x y)
}.
```

`eq` is the function that decides equality and `eqP` the theorem that insures that `(eq x y)`, written in the following as `x == y`, is true iff `x = y`. We call this the adequacy of equality.

Adding decidability on objects has the nice consequence to equate the type `bool`, the booleans, with the type `Prop`, the propositions. Of course, these two types are not identified since we are completely compatible with the standard way of doing proofs in COQ. Still, an inductive relation `reflect` of type `Prop -> bool -> Type` holds all the information to coerce one into the other.

In practice, booleans are always privileged with respect to propositions. For this, the coercion `is_true` from booleans to propositions is used.

```
Coercion is_true b := b = true.
```

As an example, let us consider equality and conjunction. Instead of stating a conjunction of two equalities as $x = y \wedge z = t$, we prefer writing it using booleans as $x == y \ \&\& \ z == t$. This simple modification gives a classical flavour to the usually intuitionistic prover COQ. Moreover, proof scripts become more similar to the ones of other systems like HOL. In particular, as booleans accommodate the substitutivity property, rewriting becomes the tactic number one. This reflection between `bool` and `Prop` is supported by the tactic language with the so-called views. As an example, consider the reflection over conjunction which is represented by the theorem `andP`

Theorem `andP`: `forall b1 b2 : bool, reflect (b1 /\ b2) (b1 && b2)`.

Suppose now that we have to prove the following goal $x == y \ \&\& \ z == t$. In order to split this goal into two subgoals, we use a combination of two tactics: `(apply/andP; split)`. The first tactic converts the `&&` into a `/\`, the second tactic can then perform the splitting. Similarly for an hypothesis, if the goal is $x == y \ \&\& \ z == t \rightarrow A$ for an arbitrary A , the tactic `(move/andP; case)` performs the conversion and the destructuring. Note that we can do even shorter combining view and case: `case/andP`.

Some standard operations are defined on `eqType`. For example, it is possible to build the set of pairs of objects. The construction is the following:

```
Structure eq_pair (d1 d2: eqType): Type := EqPair {
  eq_pi1: d1;
  eq_pi2: d2
}.

Definition pair_eq (d1 d2: eqType) (u v: eq_pair d1 d2): bool :=
  let EqPair x1 x2 := u in
  let EqPair y1 y2 := v in (x1 == y1) && (x2 == y2).
```

Once the adequacy of the equality is proved, we can build the expected type with decidable equality. This is represented by the function `prod_eqType` with the following type `prod_eqType: eqType -> eqType -> eqType`.

2.2 Sets

Sets are represented by their indicator function:

Definition `set (d: eqType) := d -> bool`.

For example, the constructor of a singleton is defined as

Definition `set1 x := fun y => (y == x)`.

A key construction is the one that allows to build a type d_1 with decidable equality from a set A whose carrier is a type d with decidable equality. This is done using the constructor `sub_eqType`:

`sub_eqType: forall d: eqType, set d -> eqType`.

d_1 is then `(sub_eqType d A)` and elements of d_1 are composed of elements of d and a proof that they belong to A .

```
Structure eq_sig (d: eqType) (A: set d): Set := EqSig {
  val: d;
  valP: A val
}.

```

Equality then only checks the first elements of the two records. As sets are represented as indicators, this equality is adequate (there is only one proof of $x = \text{true}$). Over sets, there is also the usual extensional equality, i.e. $A_1 =_1 A_2$ iff $A_1 x == A_2 x$ for all x .

2.3 Sequence

Sequences are represented in a standard way

```
Inductive seq (d: eqType): Type := Seq0 | Adds (x : d) (s : seq d).

```

Sequences are equipped with all the basic operations. In the following, we are going to use two of these operations: `size`, `count`. `size` gives the number of elements of a sequence. `count` returns the number of elements of a set inside a sequence.

2.4 Finite type

The last construction before defining groups is the one for creating finite types. A finite type is composed of a type `sort` with decidable equality, its sequence of elements and a proof that the sequence contains each element of `sort` once and only once.

```
Structure finType: Type := FinSet {
  sort :> eqType;
  enum : seq sort;
  enumP : forall x, count (set1 x) enum = 1
}.

```

Note that this encoding of finite sets gives for free an order on the elements of the finite set, i.e. the index of its occurrence in the sequence. The cardinality of a set A over a finite type S is defined as $(\text{count } A \text{ (enum } S))$. It is written in the following as $(\text{card } A)$.

3 From finite groups to Sylow's theorems

3.1 Finite group, coset and subgroup

A finite group contains a finite set, an unit element, an inverse function and a multiplication with the usual properties.

```
Structure finGroup : Type := Finite {
  element:> finType;
  unit: element;
  inv: element -> element;

```



```

    mul: element -> element -> element;
unitP: forall x, mul unit x = x;
invP:  forall x, mul (inv x) x = unit;
mulP:  forall x1 x2 x3, mul x1 (mul x2 x3) = mul (mul x1 x2) x3
}.

```

Given a multiplicative finite group G and x, y two elements of G , 1 is encoded as $(\text{unit } G)$, x^{-1} as $(\text{inv } G \ x)$, and xy as $(\text{mul } G \ x \ y)$. Given a finite group G , a set H of G and an element a of G , the left coset aH (the right coset Ha) is the set of the elements ax (respectively the set of elements xa) for all x in H . As we have x in aH iff $a^{-1}x$ is in H (respectively x in Ha iff xa^{-1} is in H), we have the following definitions:

Definition `lcoset H a`: `set G := fun x => H (a-1x)`.

Definition `rcoset H a`: `set G := fun x => H (xa-1)`.

The function $x \mapsto ax$ is a bijection between H and aH , so both sets have same cardinality. Furthermore, every coset aH can be represented by a canonical element \bar{a} such that $aH =_1 bH$ iff $\bar{a} = \bar{b}$. Technically, \bar{a} is encoded as $(\text{root } (\text{lcoset } H) \ a)$, which is the first element in the sequence of the finite set that belongs to aH .

Subgroups are not defined as structures but as sets. Their definition is a bit intricate. The idea is to say that a set H is a subgroup if it is not empty, and if x and y are in H so is xy^{-1} . This is sufficient. Since if H is non empty, it contains at least an element z , so we have $zz^{-1} = 1$ belongs to H . Also, for all x in H , $1x^{-1} = x^{-1}$ also belongs to H . Finally, if x and y belongs to H , we have y^{-1} belongs to H , so is $x(y^{-1})^{-1} = xy$. In our definition, 1 is used as a witness of non-emptiness. For the second condition, we rewrite it as “if x is in H then H is included in Hx ”.

Definition `subgrp H` :=

```

H 1 && subset H (fun x => subset H (rcoset H x)).

```

where $(\text{subset } H_1 \ H_2)$ is true iff for all x in H_1 , x is also in H_2 . In this definition, G is given implicitly since the type of H is $(\text{set } G)$. This definition is of little use for proving that a set is a subgroup. As we are in a finite setting, a much more practical characterisation of a subgroup is that it is a non-empty set that is stable by multiplication. This is represented in our development by the theorem `finstbl_sbgrp`:

Lemma `finstbl_sbgrp`: `forall G (H : set G) (a : G),`

```

H a -> (forall x y, H x -> H y -> H (xy)) -> subgrp H.

```

If H is a subgroup, its left cosets partition G : if z is in the intersection aH and bH , there exist h_1 and h_2 such that $ah_1 = z = bh_2$, we get $a = b(h_2h_1^{-1})$ and $b = a(h_1h_2^{-1})$, so $aH =_1 bH$. We denote $(\text{lindex } H)$ the number of canonical elements. We then get that `card G = lindex H * card H`. As in our development groups and subgroups differ in nature, groups hold the carrier while subgroups are only indicators, it is preferable to state Lagrange’s theorem at the level of subgroups:

Theorem `lLaGrange`:

```

forall G (H K: set G),

```

```

subgrp H -> subgrp K -> subset H K => card H * lindex H K = card K.

```

Now, $\text{index } H \ K$ denotes the number of coset of H with respect to K . Note that we can always get back to the usual statement, using the fact that G is a subgroup of itself.

3.2 Conjugate, normaliser and normal subgroup

Normal subgroups are needed for the proof of Sylow's theorem. In order to define them, we first define the conjugate operation.

Definition $y^x := x^{-1}yx$.

Then, given an arbitrary element x and an arbitrary set H the conjugate set xHx^{-1} is defined as follows:

Definition $\text{conjsg } H \ x := \text{fun } y \Rightarrow H \ y^x$.

y is in xHx^{-1} iff $x^{-1}yx$ is in H . We are now ready to define the notion of normal subgroup. H is normal in K iff for all element x in K , $xHx^{-1} =_1 H$. It is in fact sufficient to require that H is included in xHx^{-1} as both sets have same cardinality. This gives the following definition:

Definition $\text{normal } H \ K := \text{subset } K \ (\text{fun } x \Rightarrow \text{subset } H \ (\text{conjsg } H \ x))$.

Later in the proof of the first Sylow's theorem we use the property that the quotient of a group by a normal subgroup is a group. This is a direct consequence of normality that imposes that the operation of the group behaves well with respect to cosets. The quotient group is represented in our development by the group RG composed with the roots of G with respect to the left coset relation.

Given a subgroup H , it is possible to build its normaliser, the set of all x in K such that $xHx^{-1} = H$ as:

Definition $\text{normaliser } H \ K \ x :=$
 $(\text{subset } K \ (\text{fun } z \Rightarrow (\text{conjsg } H \ x \ z == H \ z))) \ \&\& \ K \ x$.

By definition, we have that H is normal in $(\text{normaliser } H \ K)$. This is the theorem `normaliser_normal`:

Lemma `normaliser_normal`:
 $\text{forall } G \ (H \ K : \text{set } G), \text{subset } H \ K \rightarrow \text{normal } H \ (\text{normaliser } H \ K)$.

3.3 Group actions

Group actions are the key construction for our final theorems. To define an action, we need a group G , a subgroup H and a finite set S . This is written in our development as:

Variable $G : \text{finGroup}$.
Variable $H : \text{set } G$.
Hypothesis `sgrp_H`: $\text{subgrp } H$.
Variable $S : \text{finType}$.

An action `to` is a homomorphism from H to the permutations of S (the bijections from S to S). This is defined as:

Variable to: $G \rightarrow (S \rightarrow S)$.
 Hypothesis to_bij: forall x, H x \rightarrow bijective (to x).
 Hypothesis to_morph: forall (x y: G) z,
 H x \rightarrow H y \rightarrow to (xy) z = to x (to y z).

where the predicate `bijective` indicates that the function is a bijection. Note that we have arbitrarily chosen to define our action `to` on G and only require the properties of homomorphism and permutation to hold for elements of H .

For an element a of S , we define its orbit as all the elements of S that can be reached from a by the function `to`. In other words, it is the image of H by the function that given an x in G associates `(to x a)`.

Definition orbit a := image (fun x => to x a) H.

We can partition S using the orbits. A key property of group action comes with the notion of stabiliser. Given an element a of S , we call its stabiliser the set of all the elements x of H that leave a unchanged by the function `to x a`. Formally, this gives

Definition stabiliser a := fun x => ((to x a) == a) && (H x).

The stabiliser is clearly a subgroup of H but the key property is that the cardinal of the orbit of a and the index of the stabiliser of a are equal.

Lemma card_orbit: forall a, card (orbit a) = lindex (stabiliser a) H.

to see this we just have to notice that we have `(to x a) =d (to y a)` iff $x^{-1}y$ is in `(stabiliser a)`. For this, we write `(to y a)` as `(to x (to (x-1y) a))` and use the fact that `to` is injective.

In the particular case where H has cardinality p^α with p prime, as orbits partition S and their cardinality is an index, Lagrange's theorem gives us that these orbits are of cardinality p^β with $\beta \leq \alpha$. Now, if we collect in the set S_0 all the elements of S whose orbit has cardinality $1 = p^0$, i.e elements that are in the stabiliser of every element of H :

Definition S_0 a := subset H (stabiliser a).

we get our central lemma

Lemma mpl: (card S) % p = (card S_0) % p.

where `%` is the usual modulo operation. All the orbits of cardinality p^β with $0 < \beta \leq \alpha$ cancel out in the modulo.

3.4 Cauchy's theorem

The proof of the first Sylow theorem is an inductive proof. Cauchy's theorem solves the base case. This theorem states that if a prime p divides the cardinality of a group, then there exists a subgroup of cardinality p . More precisely, there exists an element a , such that its cyclic group, i.e. the set of all the a^i , is of cardinality p . As we did for Lagrange's, we state this theorem at the level of subgroups. We take H a subgroup of G and a prime p that divides the cardinality of H . We first consider H^{p-1} the cartesian product $\underbrace{H \times \dots \times H}_{p-1}$. An element x

of H^{p-1} is written as (h_0, \dots, h_{p-2}) . We have $(\text{card } H^{p-1}) = (\text{card } H)^{p-1}$. We define H^* a subset of H^p as the image of H^{p-1} by the function

$$(h_0, \dots, h_{p-2}) \mapsto ((\prod_{i=0}^{p-2} h_i)^{-1}, h_0, \dots, h_{p-2}).$$

Clearly, we have $(\text{card } H^*) = (\text{card } H)^{p-1}$ and every element (h_0, \dots, h_{p-1}) of H^p such that $\prod_{i=0}^{p-1} h_i = 1$ is in H^* . Now we consider the additive group \mathbb{Z}_p and the action to from \mathbb{Z}_p to H^* defined as

$$n \mapsto \{ (h_0, h_1, \dots, h_{p-1}) \mapsto (h_{(0+n)\%p}, h_{(1+n)\%p}, \dots, h_{(p-1+n)\%p}) \}$$

Now, if we look at the set S_0 of the elements of orbit with cardinality 1. We can easily prove that S_0 is composed of the elements (h, \dots, h) such that $h^p = 1$. In one direction, such elements clearly belong to S_0 since they are left unchanged by any permutation of indexes. Conversely, if an element x belongs to S_0 , in particular $(\text{to } 1 \ x)$ is equal to x . So, if we write x as (h_0, \dots, h_{p-1}) , this means (h_0, \dots, h_{p-1}) is equal to (h_1, \dots, h_0) which in turn implies that h_0 is equal to h_1 , h_1 is equal to h_2 and so on. Now, the `mpl` lemma tells us that $(\text{card } H^*) \% p = (\text{card } S_0) \% p$, but the cardinality of H^* is divisible by p so we can conclude that the cardinality of S_0 is also divisible by p . As, $p \geq 2$, this means that there exists at least one element a different from 1 in S_0 . For this element, we have $a^p = 1$. We have that the cardinality of the cyclic group of a divides p but as p is prime and a is different of 1, the cardinality of its cyclic group is then exactly p . The exact statement of Cauchy's theorem in our development is

```
Theorem cauchy: forall G, (H : set G) p,
  subgrp H -> prime p -> p | (card h) ->
  exists a, H a && (card (cyclic a) == p).
```

where `|` denotes the divisibility and `cyclic` builds the cyclic group of an element.

3.5 Sylow's theorems

The first Sylow theorem tells us that if G is a group and K is a subgroup of G of cardinality p^n with p prime and p, n relatively prime, then there exists a subgroup of K of cardinality p^n . Such a subgroup of maximal cardinality in p is called a Sylow p subgroup. It is defined in our development as

```
Definition sylow K p H:=
  subgrpb H && subset H K && card H == expn p (dlogn p (card K)).
```

where `expn` is the exponential function and `dlogn` is the divisor logarithm, i.e $(\text{dlogn } p \ u)$ is the maximal power of p that divides u .

The proof of the first Sylow theorem is done by induction. We are going to prove that for all $i, 0 < i \leq n$, there exists a subgroup of cardinality p^i . For $i = 1$, the existence is given by Cauchy's theorem. Now, suppose that there exists a subgroup H of cardinality p^i , we are going to prove that there exists a subgroup L of cardinality p^{i+1} . We are acting by left translation with H on the left cosets of H with respect to K as follows:

$$x \mapsto \{ yH \mapsto (xy)H \}$$

The `mpl` lemma gives us $(\text{card } S_0) \% p = (\text{lindex } H \ K) \% p$. But by Lagrange's theorem we know that $(\text{lindex } H \ K)$ is equal to p^{n-i} . As $i < n$, we can conclude that the cardinal of S_0 is divisible by p . Now, if we look at the cosets that are in S_0 . They are the yH such that

$(xy)H = yH$ for all x in H . This corresponds to $y^{-1}Hy = H$ so y is in $(\text{normaliser } H \ K)$. So, we can deduce that $(\text{card } S_0) = (\text{lindex } H \ (\text{normaliser } H \ K))$. This means that if we take the quotient of the normaliser $(\text{normaliser } H \ K)$ by H , this is a group (H is normal in its normaliser) and its cardinality which is $(\text{lindex } H \ (\text{normaliser } H \ K))$ is divisible by p . We can then apply Cauchy's theorem and get the existence of a subgroup L_1 of cardinality p in the quotient. Taking the inverse image of L_1 by the quotient operation, we get a subgroup L of G whose cardinality is $\text{card } L_1 * \text{card } H = p p^i = p^{i+1}$. This ends the proof of the first Sylow theorem. The exact formal statement of this theorem is the following:

```
Theorem sylow1_cor: forall G (K: set G) p,
  subgrp K -> prime p -> 0 < dlogn p (card K) ->
  exists H : set G, sylow K p H.
```

The second Sylow theorem says that two Sylow p subgroups L_1 and L_2 of K are conjugate. For the proof, we act by left translation with L_2 on the left coset of L_1 . By the `mpl` lemma, we know the $(\text{card } S_0) \% p = (\text{lindex } L_1 \ K) \% p$. As L_1 is a Sylow p group, we have by Lagrange's theorem that $(\text{lindex } L_1 \ K)$ is equal to s , so is not divisible by p . This means that $(\text{card } S_0)$ is not divisible by p , so there exists an x in K such that xL_1 is in S_0 . But for this x , we know that for all y in L_2 , $(yx)L_1 = xL_1$, this means that L_2 is included in xL_1x^{-1} . As both sets have same cardinality, we have $L_2 =_1 xL_1x^{-1}$. The exact formal statement of this theorem is the following:

```
Theorem sylow2_cor: forall G (K: set G) p L1 L2,
  subgrp K -> prime p -> 0 < dlogn p (card K) ->
  sylow K p L1 -> sylow K p L2 ->
  exists x : G, K x /\ L2 =_1 conjsg L1 x.
```

The third Sylow theorem gives an indication on the number of Sylow p groups. It says that this number divides the cardinality of K and is equal to 1 modulo p . In order to count the number of Sylow p subgroup, we have to define the sylow subset of the power set of G as:

```
Definition syset K p := fun (H: powerSet G) => sylow K p (subdE H).
```

Now, the first part of the third theorem that regards divisibility is proved acting with K on $(\text{syset } K \ p)$ as follows:

$$x \mapsto \{ L \mapsto xLx^{-1} \}$$

The second theorem tells us that all the elements of $(\text{syset } K \ p)$ are conjugate. So, from one Sylow p subgroup L we can reach any other by conjugation. This means that $(\text{syset } K \ p)$ contains one single orbit. So, $(\text{card } (\text{syset } K \ p)) = (\text{card } (\text{orbit } L))$. The theorem `card_orbit` tells us the `card (orbit L)` is equal to $(\text{lindex } (\text{stabiliser } L) \ K)$. Using Lagrange's theorem, we get that it divides $(\text{card } K)$. The formal statement of the first part of the third Sylow theorem is the following:

```
Theorem sylow3_div: forall G (K: set G) p,
  subgrp K -> prime p -> 0 < dlogn p (card k) ->
  (card (syset K p)) | (card K).
```

For the second part, we consider H a Sylow p group for K . We act with H on $(\text{syset } K \ p)$ by conjugation as before:

$$x \mapsto \{ L \mapsto xLx^{-1} \}$$

An element L is in S_0 if $xLx^{-1} =_1 L$ for all x in H . This means that H is included in $(\text{normaliser } L \ K)$. As we have $(\text{syLOW } K \ p \ H)$, we have also $(\text{syLOW } (\text{normaliser } L \ K) \ p \ H)$. This holds also for L , so we have $(\text{syLOW } (\text{normaliser } L \ K) \ p \ L)$. The second theorem tells us that H and L are then conjugate in $(\text{normaliser } L \ K)$. But as L is normal in its normaliser, this implies that $H =_1 L$. So $(\text{card } S_0)$ is equal to 1. If we apply the `mpl` lemma we get the expected result. The formal statement of the second part of the third Sylow theorem is the following:

```
Theorem sylow3_mod: forall G (K: set G) p,
  subgrp K -> prime p -> 0 < dlogn p (card k) ->
  (card (syset K p)) % p = 1.
```

4 Conclusion

Formalising Sylow's theorems has been surprisingly smooth. One reason has to do with the fact that we have built our development on top of `SSREFLECT`. This base was used by Georges Gonthier for his proof of the four colour theorem. It has already been tested on a large development, so it is quite complete. The only basic construction we had to add is the power set. Another reason that made our life simpler is that we were working in a decidable fragment of the COQ logic. No philosophical issue about constructiveness slowed down our formalisation. Finally, Gregory Constantine's proof was perfect for our formalisation work. The only part of the formalisation that was ad-hoc was the construction of the set H^* . It represents only 360 lines of the 3550 lines of the formalisation. The fact that this experiment was positive is clearly a good sign for further formalisations in group theory.

References

- [1] Gregory M. Constantine. Group Theory. Available at <http://www.pitt.edu/~gmc/algsyl.html>.
- [2] Georges Gonthier. Notation of the Four Colour Theorem proof. Available at <http://research.microsoft.com/~gonthier/4colnotations.pdf>.
- [3] Florian Kammüller and Lawrence C. Paulson. A Formal Proof of Sylow's Theorem. *Journal of Automating Reasoning*, 23(3-4):235–264, 1999.
- [4] Freek Wiedijk. Formalizing 100 theorems. Available at <http://www.cs.ru.nl/~freek/100/>.
- [5] Helmut Wielandt. Ein beweis für die existenz der sylowgruppen. *Archiv der Mathematik*, 10:401–402, 1959.

Module groups

```

Structure finGroup: Type:= Finite {
  element:> finType;
  unit    : element;
  inv     : element → element;
  mul     : element → element → element;
  unitP   : ∀x, mul unit x = x;
  invP    : ∀x, mul (inv x) x = unit;
  mulP    : ∀x1 x2 x3, mul x1 (mul x2 x3) = mul (mul x1 x2) x3
}.

```

Section *GroupIdentities*.

Variable G : *finGroup*.

Lemma *mulgA*: $\forall x_1 x_2 x_3: G, x_1 \times (x_2 \times x_3) = x_1 \times x_2 \times x_3$.

Lemma *mul1g*: $\forall x: G, 1 \times x = x$.

Lemma *mulVg*: $\forall x: G, x^{-1} \times x = 1$.

Lemma *mulg_inv1*: $\forall x: G, \text{cancel} (\text{mulg } x) (\text{mulg } x^{-1})$.

Lemma *mulg_injl*: $\forall x: G, \text{injective} (\text{mulg } x)$.

Lemma *mulg1*: $\forall x: G, x \times 1 = x$.

Lemma *invg1*: $1^{-1} = 1$.

Lemma *mulgV*: $\forall x: G, x \times x^{-1} = 1$.

Lemma *mulg_invr*: $\forall x: G, \text{monic} (\text{mulgr } x) (\text{mulgr } x^{-1})$.

Lemma *mulg_injr*: $\forall x: G, \text{injective} (\text{mulgr } x)$.

Lemma *invg_inv*: *monic invg invg*.

Lemma *invg_inj*: *injective invg*.

Lemma *invg_mul*: $\forall x_1 x_2: G, (x_2 \times x_1)^{-1} = x_1^{-1} \times x_2^{-1}$.

Lemma *mulVg_inv1*: $\forall x: G, \text{monic} (\text{mulg } x^{-1}) (\text{mulg } x)$.

Lemma *mulVg_invr*: $\forall x, \text{monic} (\text{mulgr } x^{-1}) (\text{mulgr } x)$.

Theorem *mulg_s1*: $\forall a b:G, (b \times a^{-1}) \times a = b$.

Theorem *mulg_s2*: $\forall a b:G, (b \times a) \times a^{-1} = b$.

End *GroupIdentities*.

Definition *conjug* (G : *finGroup*) ($x y$: G):= $x^{-1} \times y \times x$.

Section *Conjugation*.

Variable G : *finGroup*.

Lemma *conjgE*: $\forall x y: G, x^y = y^{-1} \times x \times y$.

Lemma *conjg1*: *conjg 1 =₁ id*.

Lemma *conjg1g*: $\forall x: G, 1^x = 1$.

Lemma *conjg_mul*: $\forall x_1 x_2 y: G, (x_1 \times x_2)^y = x_1^y \times x_2^y$.

Lemma *conjg_invg*: $\forall x y: G, (x^{-1})^y = (x^y)^{-1}$.

Lemma *conjg_conj*: $\forall x y_1 y_2: G, (x^{y_1})^{y_2} = x^{y_1 \times y_2}$.

Lemma *conjg_inv*: $\forall y: G, \text{monic } (\text{conjg } y) (\text{conjg } y^{-1})$.

Lemma *conjg_invV*: $\forall y: G, \text{monic } (\text{conjg } y^{-1}) (\text{conjg } y)$.

Lemma *conjg_inj*: $\forall y: G, \text{injective } (\text{conjg } y)$.

Definition *conjg_fp* ($y x: G$): $= x^y =_d x$.

Definition *commg* ($x y: G$): $= x \times y = y \times x$.

Lemma *conjg_fpP*: $\forall x y: G, \text{reflect } (\text{commg } x y) (\text{conjg_fp } y x)$.

Lemma *conjg_fp_sym*: $\forall x y: G, \text{conjg_fp } x y = \text{conjg_fp } y x$.

End *Conjugation*.

Section *SubGroup*.

Variables (G : *finGroup*) (H : *set G*).

Definition *lcoset* x : *set G* := *fun y* $\Rightarrow H (x^{-1} \times y)$.

Definition *rcoset* x : *set G* := *fun y* $\Rightarrow H (y \times x^{-1})$.

Definition *subgrpb* := $H 1 \ \&\& \ \text{subset } H (\text{fun } x \Rightarrow \text{subset } H (\text{rcoset } x))$.

Definition *subgrp*: *Prop* := *subgrpb*.

Lemma *subgrpP*: *reflect* ($H 1 \wedge \forall x y, H x \rightarrow H y \rightarrow \text{rcoset } x y$) *subgrpb*.

Hypothesis *Hh*: *subgrp*.

Lemma *subgrp1*: $H 1$.

Lemma *subgrpV*: $\forall x, H x \rightarrow H x^{-1}$.

Lemma *subgrpM*: $\forall x y, H x \rightarrow H y \rightarrow H (x \times y)$.

Lemma *subgrpMl*: $\forall x y, H x \rightarrow H (x \times y) = H y$.

Lemma *subgrpMr*: $\forall x y, H x \rightarrow H (y \times x) = H y$.

Lemma *subgrpVl*: $\forall x, H \ x^{-1} \rightarrow H \ x$.

Definition *subFinGroup*: *finGroup*.

End *SubGroup*.

Lemma *subgrp_of_group*: $\forall G: \text{finGroup}, \text{subgrp } G$.

Coercion *subgrp_of_group*: *finGroup* \gg *subgrp*.

Section *LaGrange*.

Variables ($G: \text{finGroup}$) ($H: \text{set } G$).

Hypothesis ($Hh: \text{subgrp } H$).

Lemma *rcoset_refl*: $\forall x, \text{rcoset } H \ x \ x$.

Lemma *rcoset_sym*: $\forall x \ y, \text{rcoset } H \ x \ y = \text{rcoset } H \ y \ x$.

Lemma *rcoset_trans*: $\forall x \ y, \text{connect } (\text{rcoset } H) \ x \ y = \text{rcoset } H \ x \ y$.

Lemma *rcoset_csym*: *connect_sym* (*rcoset* H).

Lemma *rcoset1*: *rcoset* $H \ 1 =_1 H$.

Lemma *card_rcoset*: $\forall x, \text{card } (\text{rcoset } H \ x) = \text{card } H$.

Definition *rindex*: $= \text{n_comp } (\text{rcoset } H)$.

Theorem *rLaGrange*: $\forall K: \text{set } G,$
 $\text{subgrp } K \rightarrow \text{subset } H \ K \rightarrow \text{card } H \times \text{rindex } K = \text{card } K$.

Theorem *sugrp_divn*: $\forall K: \text{set } G,$
 $\text{subgrp } K \rightarrow \text{subset } H \ K \rightarrow \text{card } H \mid \text{card } K$.

Lemma *lcoset_refl*: $\forall x, \text{lcoset } H \ x \ x$.

Lemma *lcoset_sym*: $\forall x \ y, \text{lcoset } H \ x \ y = \text{lcoset } H \ y \ x$.

Lemma *lcoset_trans*: $\forall x \ y, \text{connect } (\text{lcoset } H) \ x \ y = \text{lcoset } H \ x \ y$.

Lemma *lcoset_csym*: *connect_sym* (*lcoset* H).

Lemma *lcoset1*: *lcoset* $H \ 1 =_1 H$.

Lemma *card_lcoset*: $\forall x, \text{card } (\text{lcoset } H \ x) = \text{card } H$.

Definition *lindex*: $= \text{n_comp } (\text{lcoset } H)$.

Theorem *lLaGrange*: $\forall K: \text{set } G,$
 $\text{subgrp } K \rightarrow \text{subset } H \ K \rightarrow \text{card } H \times \text{lindex } K = \text{card } K$.

End *LaGrange*.

Section *FinPart*.

Variables ($G: \text{finGroup}$) ($H: \text{set } G$) ($a: G$).

Hypothesis Ha : $H a$.

Hypothesis $Hstable$: $\forall x y, H x \rightarrow H y \rightarrow H (x \times y)$.

Lemma $heqah$: $(\text{lcset } H a) =_1 H$.

Lemma $heqxh$: $\forall x, H x \rightarrow (\text{lcset } H x) =_1 H$.

Lemma $heqhx$: $\forall x, H x \rightarrow (\text{rcset } H x) =_1 H$.

Lemma finstbl_sbgrp1 : $H 1$.

Lemma finstbl_mulV : $\forall x, H x \rightarrow H x^{-1}$.

Lemma finstbl_sbgrp : $\text{subgrp } H$.

End FinPart .

Section Eq .

Variable $G: \text{finGroup}$.

Theorem eq_subgroup : $\forall a b: \text{set } G, a =_1 b \rightarrow \text{subgrpb } a = \text{subgrpb } b$.

End Eq .

Section SubProd .

Variable $G: \text{finGroup}$.

Section SubProd_subgrp .

Variables ($H K: \text{set } G$).

Hypothesis $h_subgroup$: $\text{subgrp } H$.

Hypothesis $k_subgroup$: $\text{subgrp } K$.

Lemma subprod_sbgrp : $\text{prod } H K =_1 \text{prod } K H \rightarrow \text{subgrp } (\text{prod } H K)$.

Lemma sbgrp_subprod : $\text{subgrp } (\text{prod } H K) \rightarrow \text{prod } H K =_1 \text{prod } K H$.

End SubProd_subgrp .

Variables ($H K: \text{set } G$).

Hypothesis $h_subgroup$: $\text{subgrp } H$.

Hypothesis $k_subgroup$: $\text{subgrp } K$.

Lemma sbgrphk_sbgrpkh : $\text{subgrpb } (\text{prod } H K) = \text{subgrpb } (\text{prod } K H)$.

End SubProd .

Module action

Section Action .

Variable $(G: \text{finGroup}) (H: \text{set } G)$.

Hypothesis $\text{sgrp_h}: \text{subgrp } H$.

Variable $s: \text{finType}$.

Variable $\text{to}: G \rightarrow (s \rightarrow s)$.

Hypothesis $\text{to_bij}: \forall x, H \ x \rightarrow \text{bijective } (\text{to } x)$.

Hypothesis $\text{to_morph}: \forall (x \ y: G) \ z,$
 $H \ x \rightarrow H \ y \rightarrow \text{to } (x \times y) \ z = \text{to } x \ (\text{to } y \ z)$.

Theorem $\text{to_1}: \forall x, \text{to } 1 \ x = x$.

Definition $\text{stabiliser } a := \text{setI } (\text{fun } x \Rightarrow ((\text{to } x \ a) =_d \ a)) \ H$.

Definition $\text{orbit } a := \text{image } (\text{fun } z \Rightarrow \text{to } z \ a) \ H$.

Theorem $\text{orbit_to}: \forall a \ x, H \ x \rightarrow \text{orbit } a \ (\text{to } x \ a)$.

Lemma $\text{orbit_refl}: \forall x, \text{orbit } x \ x$.

Lemma $\text{orbit_sym}: \forall x \ y, \text{orbit } x \ y = \text{orbit } y \ x$.

Lemma $\text{orbit_trans}: \forall x \ y, \text{connect } \text{orbit } x \ y = \text{orbit } x \ y$.

Lemma $\text{orbit_csym}: \text{connect_sym } \text{orbit}$.

Definition $S_0 \ a := \text{subset } H \ (\text{stabiliser } a)$.

Theorem $S_0P: \forall a, \text{reflect } (\text{orbit } a =_1 \ \text{setI } a) \ (S_0 \ a)$.

Theorem $\text{stab_1}: \forall a, \text{stabiliser } a \ 1$.

Theorem $\text{subgr_stab}: \forall a, \text{subgrp } (\text{stabiliser } a)$.

Theorem $\text{subset_stab}: \forall a, \text{subset } (\text{stabiliser } a) \ H$.

Theorem $\text{orbit_from}: \forall a \ x \ (Hx: \text{orbit } a \ x),$
 $(\text{setI } (\text{roots } (\text{lcset } (\text{stabiliser } a)))) \ H \ (\text{root } (\text{lcset } (\text{iinv1 } Hx)))$.

Theorem $\text{card_orbit}: \forall a, \text{card } (\text{orbit } a) = \text{lindex } (\text{stabiliser } a) \ H$.

Theorem $\text{card_orbit_div}: \forall a, \text{card } (\text{orbit } a) \mid \text{card } H$.

Variable $n \ p: \text{nat}$.

Hypothesis $\text{prime_p}: \text{prime } p$.

Hypothesis $\text{card_h}: \text{card } H = p^n$.

Theorem $\text{mpl}: (\text{card } s) \% p = (\text{card } S_0) \% p$.

End *Action*.

Module cyclic

Section *Phi*.

Definition *phi* $n :=$ if n is $n_1 + 1$ then $\text{card } (\text{fun } x \Rightarrow \text{coprime } n \text{ (val } x))$ else 0.

Theorem *phi_mult*: $\forall m \ n, \text{coprime } m \ n \rightarrow \text{phi } (m \times n) = \text{phi } m \times \text{phi } n$.

Theorem *phi_prime_k*: $\forall p \ k, \text{prime } p \rightarrow \text{phi } p^{k+1} = p^{k+1} - p^k$.

End *Phi*.

Section *Cyclic*.

Variable G : *fnGroup*.

Fixpoint *gexpn* $(a:G) (n: \text{nat}) \{ \text{struct } n \}$: $G :=$
if n is $n_1 + 1$ then $a \times (\text{gexpn } a \ n_1)$ else 1.

Theorem *gexpn0*: $\forall a, \text{gexpn } a \ 0 = 1$.

Theorem *gexpn1*: $\forall a, \text{gexpn } a \ 1 = a$.

Theorem *gexp1n*: $\forall n, \text{gexpn } 1 \ n = 1$.

Theorem *gexpnS*: $\forall a \ n, \text{gexpn } a \ (n + 1) = a \times \text{gexpn } a \ n$.

Theorem *gexpn_h*: $\forall n \ a \ H, \text{subgrp } H \rightarrow H \ a \rightarrow H \ (\text{gexpn } a \ n)$.

Theorem *gexpn_add*: $\forall a \ n \ m, \text{gexpn } a \ n \times \text{gexpn } a \ m = \text{gexpn } a \ (n + m)$.

Theorem *gexpn_mul*: $\forall a \ n \ m, \text{gexpn } (\text{gexpn } a \ n) \ m = \text{gexpn } a \ (n \times m)$.

Fixpoint *seq_fn* $(f: G \rightarrow G) (n: \text{nat}) (a: G) (L: \text{seq } G) \{ \text{struct } n \}$: $\text{seq } G :=$
if n is $n_1 + 1$ then
if $\text{negb } (L \ a)$ then $\text{seq_fn } f \ n_1 \ (f \ a)$ (*Adds a L*) else L else L .

Definition *seq_f* $f \ a := \text{seq_fn } f \ (\text{card } G) \ a \ (\text{Seq0 } _)$.

Definition *cyclic* $a := \text{seq_f } (\text{fun } x \Rightarrow a \times x) \ 1$.

Theorem *cyclic1*: $\forall a, \text{cyclic } a \ 1$.

Theorem *cyclicP*: $\forall a \ b, \text{reflect } (\exists n, \text{gexpn } a \ n =_d b) \ (\text{cyclic } a \ b)$.

Theorem *cyclic_h*: $\forall a \ H, \text{subgrp } H \rightarrow H \ a \rightarrow \text{subset } (\text{cyclic } a) \ H$.

Theorem *cyclic_min*: $\forall a \ b,$
 $\text{cyclic } a \ b \rightarrow \exists m, (m < \text{card } (\text{cyclic } a)) \ \&\& \ (\text{gexpn } a \ m =_d b)$.

Theorem *cyclic_in*: $\forall a \ m, \text{cyclic } a \ (\text{gexpn } a \ m)$.

Theorem *subgr_cyclic*: $\forall a, \text{subgrp } (\text{cyclic } a)$.

Theorem *cyclic_expn_card*: $\forall a, \text{gexpn } a \ (\text{card } (\text{cyclic } a)) =_d 1$.

Theorem *cyclic_div_card*: $\forall a \ n, \text{card } (\text{cyclic } a) \mid n = (\text{gexpn } a \ n =_d 1)$.

Theorem *cyclic_div_g*: $\forall a, \text{card } (\text{cyclic } a) \mid \text{card } G$.

Module normal

Section *Normal*.

Variables (G : *finGroup*) (H K : *set G*).

Hypothesis *sgrp_h*: *subgrp H*.

Hypothesis *sgrp_k*: *subgrp K*.

Hypothesis *subset_hk*: *subset H K*.

Definition *conjsg* x y : $H(y^x)$.

Theorem *conjsg1*: $\forall x$, *conjsg* x 1 .

Theorem *conjsg1g*: $\forall x$, *conjsg* 1 $x = H$ x .

Theorem *conjsg_inv*: $\forall x$ y , *conjsg* x $y \rightarrow$ *conjsg* x y^{-1} .

Theorem *conjsg_conj*: $\forall x$ y z , *conjsg* $(x \times y)$ $z =$ *conjsg* y (z^x) .

Theorem *conjsg_subgrp*: $\forall x$, *subgrp* (*conjsg* x).

Theorem *conjsg_image*: $\forall y$,
conjsg $y =_1$ *image* (*conjg* y^{-1}) H .

Theorem *conjsg_inv1*: $\forall x$,
 $($ *conjsg* $x) =_1 H \rightarrow ($ *conjsg* $x^{-1}) =_1 H$.

Theorem *conjsg_card*: $\forall x$,
card (*conjsg* x) = *card* H .

Theorem *conjsg_subset*: $\forall x$,
subset H (*conjsg* x) $\rightarrow ($ *conjsg* $x) =_1 H$.

Theorem *lcset_root*: $\forall x$, *lcset* H x (*root* (*lcset* H) x).

Definition *normalb*: $=$ *subset* K (*fun* $x \Rightarrow$ *subset* H (*conjsg* H x)).

Definition *normal*: *Prop*: $=$ *normalb*.

Hypothesis *normal_k*: *normal*.

Theorem *conjsg_normal*: $\forall x$, K $x \rightarrow$ *conjsg* $x =_1 H$.

Definition *rootSet*: $=$ *subFin* (*setI* (*roots* (*lcset* H)) K).

Theorem *card_rootSet*: *card* *rootSet* = *lindex* H K .

Theorem *unit_root_sub*:
setI (*roots* (*lcset* H)) K (*root* (*lcset* H) 1).

Definition *unit_root*: *rootSet*.

Definition *mult_root*: *rootSet* \rightarrow *rootSet* \rightarrow *rootSet*.

Definition *inv_root*: *rootSet* \rightarrow *rootSet*.

Theorem *unitP_root*: $\forall x, \text{mult_root } \text{unit_root } x = x$.

Theorem *invP_root*: $\forall x, \text{mult_root } (\text{inv_root } x) x = \text{unit_root}$.

Theorem *mulP_root*: $\forall x_1 x_2 x_3,$
 $\text{mult_root } x_1 (\text{mult_root } x_2 x_3) = \text{mult_root } (\text{mult_root } x_1 x_2) x_3$.

Definition *root_group* := (*Group.Finite* *unitP_root* *invP_root* *mulP_root*).

Theorem *card_root_group*: *card* *root_group* = *lindex* *H* *K*.

End *Normal*.

Section *NormalProp*.

Variables (*G*: *finGroup*) (*H* *K*: *set* *G*).

Hypothesis *sgrp_h*: *subgrp* *H*.

Hypothesis *sgrp_k*: *subgrp* *K*.

Hypothesis *subset_hk*: *subset* *H* *K*.

Hypothesis *normal_hk*: *normal* *H* *K*.

Theorem *normal_subset*: $\forall L,$
 $\text{subgrp } L \rightarrow \text{subset } H L \rightarrow \text{subset } L K \rightarrow \text{normal } H L$.

Definition *RG* := (*root_group* *sgrp_h* *sgrp_k* *subset_hk* *normal_hk*).

Theorem *th_quotient*: $\forall x, K x \rightarrow$
 $(\text{setI } (\text{roots } (\text{lcset } H)) K (\text{root } (\text{lcset } H) x))$.

Definition *quotient*: *G* \rightarrow *RG*.

Theorem *quotient_lcset*: $\forall x, K x \rightarrow \text{lcset } H x (\text{val } (\text{quotient } x))$.

Theorem *quotientI*: $\forall x, H x \rightarrow \text{quotient } x = 1$.

Theorem *quotient_morph*: $\forall x y,$
 $K x \rightarrow K y \rightarrow \text{quotient}(x \times y) = \text{quotient}(x) \times \text{quotient}(y)$.

Theorem *quotient_image_subgrp*: $\forall L,$
 $\text{subset } H L \rightarrow \text{subset } L K \rightarrow \text{subgrp } L \rightarrow \text{subgrp } (\text{image } \text{quotient } L)$.

Theorem *quotient_preimage_subgrp*: $\forall L,$
 $\text{subgrp } L \rightarrow \text{subgrp } (\text{setI } (\text{preimage } \text{quotient } L) K)$.

Theorem *quotient_preimage_subset_h*: $\forall L,$
 $\text{subgrp } L \rightarrow \text{subset } H (\text{setI } (\text{preimage } \text{quotient } L) K)$.

Theorem *quotient_preimage_subset_k*: $\forall L, \text{subset } (\text{setI } (\text{preimage } \text{quotient } L) K) K$.

Theorem *quotient_index*: $\forall L, \text{subset } H L \rightarrow \text{subset } L K \rightarrow \text{subgrp } L \rightarrow$
 $\text{lindex } H L = \text{card } (\text{image } \text{quotient } L)$.

Theorem *quotient_image_preimage*: $\forall L,$
image quotient (setI (preimage quotient L) K) =₁ L.

End *NormalProp*.

Section *Normalizer*.

Variables (G : *finGroup*) (H K : *set G*).

Hypothesis *sgrp_h*: *subgrp H*.

Hypothesis *sgrp_k*: *subgrp K*.

Hypothesis *subset_hk*: *subset H K*.

Definition *normaliser* $x :=$
(subset K (fun z \Rightarrow (conjsg x z =_d H z))) && K x.

Theorem *normaliser_grp*: *subgrp normaliser*.

Theorem *normaliser_subset*: *subset normaliser K*.

Theorem *subset_normaliser*: *subset H normaliser*.

Theorem *normaliser_normal*: *normal H normaliser*.

Theorem *card_normaliser*:
card (root_group sgrp_h normaliser_grp subset_normaliser
normaliser_normal) = lindex H normaliser.

End *Normalizer*.

Section *Eq*.

Variables G : *finGroup*.

Theorem *eq_conjsg*: $\forall a$ b $x, a =_1 b \rightarrow$ *conjsg a x =₁ conjsg b x.*

End *Eq*.

Section *Root*.

Variable (G : *finGroup*) (H : *set G*).

Hypothesis *sgrp_h*: *subgrp H*.

Theorem *root_lcoset1*: H (*root (lcoset H) 1*).

Theorem *root_lcosetd*: $\forall a, H$ ($a^{-1} \times$ *root (lcoset H) a*).

End *Root*.

Module leftTranslation

Section *LeftTrans*.

Variable $(G: \text{finGroup}) (H K L: \text{set } G)$.

Hypothesis *sgrp_k*: *subgrp* K .

Hypothesis *sgrp_l*: *subgrp* L .

Hypothesis *sgrp_h*: *subgrp* H .

Hypothesis *subset_hk*: *subset* $H K$.

Hypothesis *subset_lk*: *subset* $L K$.

Definition *ltrans*: $G \rightarrow \text{rootSet } L K \rightarrow \text{rootSet } L K$.

Theorem *ltrans_bij*: $\forall x, H x \rightarrow \text{bijective } (\text{ltrans } x)$.

Theorem *ltrans_morph*: $\forall x y z,$

$H x \rightarrow H y \rightarrow \text{ltrans } (x \times y) z = \text{ltrans } x (\text{ltrans } y z)$.

End *LeftTrans*.

Module sylow

Section *Cauchy*.

Variable $(G: \text{finGroup}) (H: \text{set } G)$.

Hypothesis *sgrp_h*: *subgrp* H .

Variable $p: \text{nat}$.

Hypothesis *prime_p*: *prime* p .

Hypothesis *p_divides_h*: $p \mid \text{card } H$.

Theorem *cauchy*: $\exists a, H a \ \&\& \ \text{card } (\text{cyclic } a) =_d p$.

End *Cauchy*.

Section *Sylow*.

Variable $(G: \text{finGroup}) (K: \text{set } G)$.

Hypothesis *sgrp_k*: *subgrp* K .

Variable $p: \text{nat}$.

Hypothesis *prime_p*: *prime* p .

Let $n := d \logn p (\text{card } K)$.

Hypothesis *n_pos*: $0 < n$.

Definition *sylow* $L := (\text{subgrp } L) \ \&\& \ (\text{subset } L K) \ \&\& \ (\text{card } L =_d p^n)$.

Theorem *eq_sylow*: $\forall a b, a =_1 b \rightarrow \text{sylow } a = \text{sylow } b$.

Theorem *sylow_conjsg*: $\forall L_1 x, K x \rightarrow \text{sylow } L_1 \rightarrow \text{sylow } (\text{conjsg } L_1 x)$.

Theorem *sylow1_rec*: $\forall i H i, 0 < i \rightarrow i < n \rightarrow$
 $\text{subgrp } H i \rightarrow \text{subset } H i K \rightarrow \text{card } H i = p^i \rightarrow$
 $\exists H: \text{set } G,$
 $\text{subgrp } H \wedge \text{subset } H i H \wedge \text{subset } H K \wedge \text{normal } H i H \wedge \text{card } H = p^{i+1}.$

Theorem *sylow1*: $\forall i, 0 < i \rightarrow i \leq n \rightarrow$
 $\exists H: \text{set } G, \text{subgrp } H \wedge \text{subset } H K \wedge \text{card } H = p^i.$

Theorem *sylow1_cor*: $\exists H: \text{set } G, \text{sylow } H$.

Theorem *sylow2*: $\forall H L i, 0 < i \rightarrow i \leq n \rightarrow$
 $\text{subgrp } H \rightarrow \text{subset } H K \rightarrow \text{card } H = p^i \rightarrow \text{sylow } L \rightarrow$
 $\exists x, (K x) \&\& \text{subset } H (\text{conjsg } L x).$

Theorem *sylow2_cor*: $\forall L_1 L_2, \text{sylow } L_1 \rightarrow \text{sylow } L_2 \rightarrow$
 $\exists x, (K x) \wedge (L_2 =_1 \text{conjsg } L_1 x).$

Definition *syset* $p := \text{sylow } (val p)$.

Theorem *sylow3_div*: $\text{card } \text{syset } p \mid \text{card } K$.

End *Sylow*.

Section *SylowAux*.

Variable $(G: \text{finGroup}) (H K L: \text{set } G)$.

Hypothesis *sgrp_k*: $\text{subgrp } K$.

Hypothesis *sgrp_l*: $\text{subgrp } L$.

Hypothesis *sgrp_h*: $\text{subgrp } H$.

Hypothesis *subset_hl*: $\text{subset } H L$.

Hypothesis *subset_lk*: $\text{subset } L K$.

Variable $p: \text{nat}$.

Hypothesis *prime_p*: $\text{prime } p$.

Let $n := \text{dlogn } p (\text{card } K)$.

Hypothesis *n_pos*: $0 < n$.

Theorem *sylow_subset*: $\text{sylow } K p H \rightarrow \text{sylow } L p H$.

End *SylowAux*.

Section *Sylow3*.

Variable $(G: \text{finGroup}) (K: \text{set } G)$.

Hypothesis *sgrp_k*: $\text{subgrp } K$.

Variable $p: \text{nat}$.

Hypothesis *prime_p*: *prime p*.

Let $n := d \log_n p (\text{card } K)$.

Hypothesis *n_pos*: $0 < n$.

Theorem *sylow3_mod*: $\text{card } (\text{syset } K \ p) \% p = 1$.

End *Sylow3*.



Unité de recherche INRIA Sophia Antipolis
2004, route des Lucioles - BP 93 - 06902 Sophia Antipolis Cedex (France)

Unité de recherche INRIA Futurs : Parc Club Orsay Université - ZAC des Vignes
4, rue Jacques Monod - 91893 ORSAY Cedex (France)

Unité de recherche INRIA Lorraine : LORIA, Technopôle de Nancy-Brabois - Campus scientifique
615, rue du Jardin Botanique - BP 101 - 54602 Villers-lès-Nancy Cedex (France)

Unité de recherche INRIA Rennes : IRISA, Campus universitaire de Beaulieu - 35042 Rennes Cedex (France)

Unité de recherche INRIA Rhône-Alpes : 655, avenue de l'Europe - 38334 Montbonnot Saint-Ismier (France)

Unité de recherche INRIA Rocquencourt : Domaine de Voluceau - Rocquencourt - BP 105 - 78153 Le Chesnay Cedex (France)

Éditeur
INRIA - Domaine de Voluceau - Rocquencourt, BP 105 - 78153 Le Chesnay Cedex (France)
<http://www.inria.fr>
ISSN 0249-0803